
Washa! / LED
Washa swichi ya IchigoJam. Chapa "LED1" katika kibodi yako. Bonyeza
kibonye ENTER. (Kibonye ENTER kiko upande wa kulia na ni kikubwa)

LED1


Ni sawasawa endapo LED ya IchigoJam iliwaka na ukapata ujumbe "OK".
Chapa "LED0" ili kuzima.

LED0


Kibonye SPACE ni kikubwa na kiko katikati, kwa chini katika
kibodi.

Type "LED 1" and the press ENTER key.

LED 1


Ni sawa vyovyote vile, nafasi ikiwepo au la.
Hebu chapa "RED0" na bonyeza kibonye ENTER.

RED0


LED haitazimika, utapata ujumbe "Syntax error".

Usihofu kuhusu kufanya makosa mara kwa mara! kompyuta haikasiriki kamwe.
Hebu chapa "ABD" bila kubonyeza kibonye ENTER.

ABD


Mara moja bonyeza kibonye Backspace.

Kibonye Backspace kiko juu upande wa kulia katika kibodi.

AB


Unaweza kufuta kibambo (character)! Hivyo usifofu kuhusu kukosea kuchapa.

Jaribu sasa!

1. Chapa "ABCDEFG"

2. Jaribu kuzima LED

3. Jaribu kuizima na kuiwasha kwa haraka

4. Chapa "BEEP" na bonyeza kibonye ENTER

5. Chapa "CLS" na bonyeza kibonye ENTER

Dhibiti muda / WAIT
Hebu tufanye "wait" katika IchigoJam.

Kielekezi cha mraba kinachopepesa katika skrini kina maana kuwa kibodi iko
tayari kutumika.

WAIT60


Je umeona kuwa kielekezi kimepotea kwa muda mfupi? Hebu tujaribu tena.
Kama utabadili namba iliyo mbele ya "WAIT", urefu wa muda katika "wait"
utaweza kubadilika.

※1 second = 60

WAIT120


Kielekezi kimepotea kwa muda mrefu zaidi kulinganisha na mwanzo.
Kuwasha LED, na kuzima otomatiki.

LED1:WAIT120:LED0


Tumia alama ya nukta mbili ':'(ni rahisi kuichanganya na ya nuktamkato ';')
kuunganisha komandi.

Washa kwa "LED1", subiri kwa "WAIT120", zima kwa "LED0".
Kuwasha LED kidogo.

LED1:WAIT3:LED0


Kama utabadili namba kwa ajili ya "WAIT", itabadili muda wa uwakaji wa LED.
Washa LED mara mbili.

LED1:WAIT3:LED0:WAIT3:LED1:WAIT3:LED0


Unaweza kuunganisha zaidi na zaidi kwa kutumia alama ya nukta mbili ":".

Jaribu sasa!

1. Hebu tufanye "WAIT" kwa sekunde 10

2. Washa LED kwa sekunde 5 na kuizima

3. Hebu tufanye "WAIT 10000", kuikatisha katikati tumia kibonye ESCAPE

4. Washa kwa muda mrefu na washa kidogo

5. Hebu jaribu kutengeneza mwanga wa LED katika ruwaza ya kuvutia

Jibu katika namba / INPUT
Programu ambayo inatatua hesabu mbalimbali za kujumlisha kutumia "RND"
(random).

Tumia "LET" kuhifadhi namba katika A mpaka Z.

10 LET A,RND(10)

20 LET B,RND(10)

30 ?A;"+";B

RUN


Hebu dabiri ("RUN") mara kadhaa!

Ili kuhifadhi namba kwa kutumia kibodi, tumia komandi ya "INPUT".

40 INPUT C

50 ?C

RUN


Mbele ya "?", chapa namba kwa kibodi na bonyeza kibonye ENTER.

Namba iliyowekwa katika C itahifadhiwa!
Kwa kuangalia majibu, "IF" inatumika kuamua.

60 IF C=A+B ?"SAWA"


Hebu tumia "ELSE" pale unapotaka kuonyesha makosa.

60 IF C=A+B ?"SAWA" ELSE ?"SI SAWA"


Pia inaweza kujitokeza ikiwa na swali kimaandishi.

40 INPUT "JIBU NI?",C


Jaribu sasa!

1. Hebu iunde upya iwe ya kujumlisha namba yenye tarakimu mbili.

2. Hebu iunde upya iwe hesabu rahisi mno

3. Hebu iunde upya iwe hesabu ya kuzidisha

4. Hebu pima muda wa kujibu kwa kutumia "TICK"

5. Hebu tengeneza mchezo wa kugonga namba

Kokotoa namba / PRINT
Hebu ipe swali kompyuta ambayo iko vizuri katika kukokotoa

Tumia "PRINT" kuonyesha ile iliyoainishwa.

PRINT 1+1


Bonyeza kibonye ENTER, mara-moja unaweza kupata jibu 「2」.

PRINT 15-7


Pia inawezekana kutoa.!

Katika kompyuta, alama ya kuzidisha inawakilishwa na kinyota '*'.

PRINT 3*5


Kugawanya ni mkwaju '/'. (namba ya juu ya sehemu:kushoto , namba ya chini ya
sehemu:kulia)

PRINT 10/2


Kwa kweli, badala ya "PRINT" unaweza kutumia '?'.

?10/3


Oh!! Jibu ni la Uongo?

Ni jibu sahili katika "IchigoJam" ambayo haitumii desimali.

Baki inakokotolewa kwa kutumia '%'.

?10%3


Kwa sababu 10 gawanya kwa 3 ni 3 na baki ni 1, '1' itaonyeshwa.
Hatimaye, hebu tukokotoe namba kubwa.

?1000*100


Oh, jibu ni la ajabu!.

Kwa kweli, masafa ambayo IchigoJam inaweza kukokotoa yanaamuliwa kutoka
-32768 mpaka 32767.
Jaribu sasa!

1. Hebu tutengeneze hesabu "100 + 200" katika IchigoJam

2. Hebu tutengeneze hesabu "8 * 7" katika IchigoJam

3. Hebu tutengeneze hesabu "100 / 25" katika IchigoJam

4. Hebu tukokotoe baki ya "30/8" kwa IchigoJam

5. Hebu tukokotoe 100,000,000 (10,000 * 10,000) kwa IchigoJam

Endesha mahali au sehemu / LOCATE
Hebu onyesha katika sehemu iliyoamuliwa!

Tumia "LOCATE" kuamua sehemu.

LOCATE 16,12:PRINT "*"


Bonyeza kibonye ENTER, '*' itaonyeshwa karibia na katikati.

Inamaanisha kwamba '*' itaonyeshwa katika nafasi ya 16 tokea upande wa
kushoto, na nafasi ya 12 tokea upande wa juu.
Hebu tuonyeshe katika sehemu ya tofauti.

LOCATE 25,5:PRINT "*"


Badili namba katika "LOCATE", eneo la uonyesho litabadilika.
Kwa kweli, unaweza kutumia "LC" kama kifupisho cha "LOCATE".

LC 16,14:PRINT "*"


Pia vibambo vinaweza kuonyeshwa katika sehemu mbili tofauti.

LC 16,12:PRINT "@":LC 10,10:PRINT "*"


Pia jibu lililokokotolewa linaweza kuonyeshwa.

LC 16,14:PRINT 2+3


Unapotaka kufuta monita, bonyeza kibonye F1 au "CLS".

CLS


Jaribu sasa!

1. Onyesha "IchigoJam" katika 20 tokea kushoto na 10 tokea upande wa juu

2. Onyesha '@' katika sehemu mbili tofauti, yaani, katika 10 tokea upande wa
kushoto, 17 tokea upande wa juu, na katika 25 tokea upande wa kushoto, 18
tokea upande wa juu

3. Onyesha matokeo ya "10-5" katika 15 tokea upande wa kushoto, 24 tokea
upande wa juu

4. Onyesha kibambo ukipendacho katika sehemu uipendayo

5. Onyesha matokeo ya mahesabu uyapendayo katika sehemu uipendayo

Iache iamue / IF
"IF" ni komandi ya kuacha kompyuta iamue.


10 IF BTN()=0 GOTO 10

RUN


kama kitufe hakijabonyezwa, mstari wa 10 utajirudia.

Pale utakapokibonyeza kitufe, programu itaenda kwa unaofuata, na kumalizika.
Kila wakati bonyeza kitufe, LED iwake na kuzima.

10 IF BTN()=0 GOTO 10

20 LED1

30 IF BTN()=0 GOTO 30

40 LED0

50 GOTO 10


Oh, wakati unabonyeza kitufe, LED inawaka?

Kwa kuwa kompyuta ina kasi, unapobonyeza kitufe itawasha LED na kusogea
katika maamuzi mengine maramoja.
Ili isiende katika maamuzi mengine maramoja, ongezea subiri "WAIT".

10 IF BTN()=0 GOTO 10

20 LED1:WAIT30

30 IF BTN()=0 GOTO 30

40 LED0:WAIT30

50 GOTO 10


Jaribu sasa!

1. Endelea tu kubonyeza kitufe

2. Jaribu kubadili "WAIT30" kuwa "WAIT60"

3. Jaribu kubadili "WAIT30" kuwa "WAIT5"

4. Jaribu kubadili "IF BTN()=1"

5. Tengeneza programu ambayo kitufe kitapobonyezwa itaanza kupepesa LED

6. Tengeneza programu ambayo itaendesha LED katika hali ya kufurahisha

Tengeneza ishara ya usiku / GOTO
Ishara yenye kupepesa kwa ajili ya usiku, inawaka na kuzima kila sekunde 0.5.

10 LED1:WAIT30:LED0:WAIT30

20 LED1:WAIT30:LED0:WAIT30

RUN


Chapa mstari wa 10 na bonyeza kibonye ENTER, kisha rejesha kielekezi mstari
wa 10 na badili kiongozo "10" kuwa "20", na bonyeza kibonye ENTER.

Hakiki kama mstari wa 10 umenakiliwa katika mstari wa 20 kwa kutumia "LIST"!
Pepesa muda wote kwa "GOTO". Takribani mara 30,000 mpaka kunakucha.

10 LED1:WAIT30:LED0:WAIT30

20 GOTO 10


Namba iliyo baada ya "GOTO" inatoa muelekeo wa fikio.

> mstari wa 10, pepesa mara moja

> mstari 20, rukia mstari wa 10

Kuisimamisha, bonyeza kibonye [ESC].
Pepesa kwa haraka

LIST

10 LED1:WAIT3:LED0:WAIT3


Tengeneza mstari wa 15 kati ya mstari wa 10 na wa 20, kwa kubadilisha kiongozo
"10" kuwa 15 na wait3 kuwa wait30, bonyeza kibonye ENTER.
Onyesha
programu kwa "LIST"(ama kibonye =F4), na kisha tekeleza kwa "RUN"(ama
kibonye =F5).


15 LED1:WAIT30:LED0:WAIT30


Rudia kwa kubadilishana kati ya pepesa fupi na pepesa refu.
Jaribu sasa!

1. Hebu turejee katika programu ya ishara ya usiku

2. Refusha muda wa kuwaka pekee

3. Jaribu kuitengeneza iwe mchezo wa kupepesa kwa kasi na isimame ambapo
inawaka

4. Tengeneza mistari ya namba, kuanzia 100 kwa hatua 100 m.f. 100,200,300

5. Hebu tutengeneze ruwaza ya kupepesa yenye kuvutia, na ringishia kwa yoyote.

Tunza programu / SAVE/LOAD/NEW
Ni programu ya kuwasha LED. Tunza programu kwa kutumia "SAVE".

SAVE0

Saved 30byte

OK


Endapo uwezo na "OK" vitaonyeshwa, kuhifadhi programu kumemalizika.

IchigoJam inaweza kuhifadhi programu 4.(0～3)

Hebu tuzime umeme na kupakua programu kwa kutumia "LOAD".

LOAD0

Loaded 30byte

OK


Hakiki kwa kutumia "LIST"! Tekeleza kwa kutumia "RUN"!
Kuhakiki programu iliyotunzwa, tumia "FILES".

FILES


Kuwa makini usije ukafuta programu uipendayo kwa makosa.
Kutengeneza programu mpya, tumia "NEW".

NEW

OK

LIST

OK


Kama utachapa "LIST", hakuna chochote kitakachoonyeshwa.
Programu iliyotunzwa katika namba 0 itaanza otomatiki endapo utawasha swichi
wakati umeshikilia kitufe cha IchigoJam.

(Hata kama IchigoJam itatenganishwa na kibodi au scrini, itaweza kutumika kwa
operesheni za kielektroniki n.k.)

Jaribu sasa!

1. Hebu tunza programu ambayo itawasha mara 3

2. Hebu fanya ianze otomatiki

3. Angalia nini kitatokea endapo utatumia "SAVE" baada ya "NEW"

Kumbumbu ya programu / RUN/LIST
Hebu tujaribu kuhifadhi ruwaza ya kupepesa ya LED.

10 LED1:WAIT10


Kwa kuanza na namba, katika komandi inamaanisha "kuhifadhi".

Lakini LED haijawaka.

Komandi ya kudabiri ni "RUN" na kibonye ENTER, LED itawaka.

RUN


Inayofuata inaanza na 20, na ongeza kumbukumbu kwa kibonye ENTER.

20 LED0:WAIT10


Tumia "RUN" kudabiri programu iliyohifadhiwa.

Washa tena kwa "RUN"!

Ita kumbukumbu kwa "LIST".

LIST

10 LED1:WAIT10

20 LED0:WAIT10


Kupachika "WAIT" kati ya mstari wa 10 na 20.

15 WAIT50


Hakiki kwa "LIST", tekeleza kwa "RUN"!
Programu zenye namba za kufanana zitapandana (overwritten).

(Hakiki kwa "LIST"!)

15 WAIT110


Chapa namba pekee na bonyeza ENTER, itaufuta mstari ule.

15

LIST


Jaribu sasa!

1. Tumia kibonye F5 badala ya komandi ya "RUN"

2. Tumia kibonye F4 badala ya "LIST"
3. Hebu tutengeneze program itakayowasha LED mara mbili
4. Hebu tutengeneze programu itakayowasha LED kwa ruwaza ya kuvutia

Tengeneza sauti / BEEP
Hakikisha miguu ya saketi ya sauti imechomekwa katika SOUND(SND) na GND
kwenye IchigoJam yako.

BEEP


Je umesikia mlio wa sauti beep?
Toni ya juu.

BEEP 5


Toni ya chini.

BEEP 20


Toni ndefu.

BEEP 10,30


Hebu tutengeneze mdundo!

10 BEEP 5:WAIT 30

20 BEEP 20:WAIT 30

30 GOTO 10

RUN


WAIT ni komandi ya kusubiri. Mdundo wa taratibu utatokea kama utaweka namba
kubwa zaidi. GOTO ni komandi ya kurudi nyuma katika mstari wa 10 ili kurudia.
Bonyeza kibonye [ESC] kusimamisha.
Mdundo utasimama wakati ukibonyeza kitufe cha IchigoJam yako.


15 IF BTN() GOTO 15


Hebu Jaribu!
1. Tengeneza sauti mbalimbali!

2. Tengeneza mdundo wenye tempo ya kasi (fast tempo)

3. Ongeza programu hii "12 BEEP 20:WAIT 30" ili kubadili iwe tripple time
(musical time pattern)

4. Ifanye ya mdundo wa quadruple

5. Badili mdundo "GOTO 15" kuwa "GOTO 10" katika mstari wa 15

6. Tengeneza mdundo wako asili!

Tunga muziki / PLAY
Hakikisha miguu ya saketi ya sauti imechomekwa katika SOUND(SND) na GND
kwenye IchigoJam yako.

PLAY"CDE"


Je umesikia sauti Do-re-mi?
Toni za Do-re-mi-fa-so-la-ti zinawakilishwa na CDEFGAB

PLAY"CDEFGAB"


Oktavo (=kiwango cha sauti) inaweza kubadilishwa kwa kutumia alfabeti "O"
ikifuatiwa na namba.

PLAY"O4 CDEFGAB O5 C"


Unaweza kubadili urefu wa toni kwa kuweka namba.

PLAY"CDE2"


"2" katika "CDE2" inamaanisha nusu noti (half note). Mpangilio wa chaguo-msingi
(bila namba) ni robo noti. Kwa kuweka "8" inamaanisha noti ya nane ambayo
urefu wa toni yake ni nusu ya robo noti. "1" ina maana ya noti nzima.

"#" (sharp) Ni kuongeza toni kwa nusu toni, "-" (toa) ni kupunguza toni kwa nusu
toni.


PLAY"C C# F-"


"R" inamaanisha unyamavu. Unaweza kubadili urefu wa unyamavu wa toni kwa
kuweka namba baada ya R mfano "R2".

PLAY"CRC"


"T" inamaanisha kuiaandaa tempo. Mpangilio wa chaguo-msingi ni "T120"
ikimaanisha robo noti 120 kwa sekunde.

PLAY"T240 CDE2 CDE2 GEDCDED2"


Hebu tujaribu!

1. Tengeneza muziki "Tulip" "Twinkle-Twinkle" "Old-MacDonald"

2. Chapa "PLAY" tu wakati wa kucheza muziki

3. Tengeneza muziki "wimbo wa Froggie"

4. Hebu tucheze kwa pamoja kama okestra na marafiki zako!

5. Hebu tunga muziki kutoka katika kitabu chako cha muziki!

Kuzidisha kwa rejeleza FOR/NEXT
10 A=5

20 FOR I=1 TO A:?"*";:NEXT:?


Unaweza kuhesabu muundo-wa-herufi I kutokea 1 mpaka A kwa komandi ya
FOR/NEXT. Unaweza kupata alama 5 za kinyota '*'.
Hebu igeuze itumie namba uipendayo kwa komandi ya INPUT.

10 INPUT A


Chapa namba zozote!

Ongeza namba na rejeleza!

15 INPUT B

17 FOR J=1 TO B

30 NEXT


Unaweza kuzidisha!
Hesabu muundo-wa-herufi C ambayo inahesabu idadi ya alama za kinyota '*'.

16 C=0

25 C=C+A

40 ?C


C=C+A kwa kuwa unaweza kupata alama A x '*' kila rejeleza. Kutumia LET C,C+1
ni SAWA pia. Kama hutafanya ingizo C=0 au LET C,0, basi C itaongezeka katika
kila utekelezaji mfululizo.
Hebu jaribu

1. Kokotoa 7 zidisha kwa 8

2. Kokotoa 8 zidisha kwa 7

3. Badili kuwa alama uipendayo

4. Angalia ili ubadili alama kuwa ya herufi 2

5. Angalia kama ingizo ni sifuri

Tengenza namba jozi / BIN$
Kuna namba, silabi, alfabeti, vibambo mbalimbali, lakini kibambo kinachotumika
na kompyuta ni "0" na "1" tu.

Kwa kawaida, kama ukijumlisha 1 katika 9, itakuwa 10 na vibambo 2, lakini
kwasababu kuna 0 na 1 tu, moja pekee itajumlishwa kwa 1, hivyo itakuwa
10(moja sifuri) na vibambo 2.
Hii ni namba jozi.

"BIN$" ni komandi inayobadili namba kuwa namba jozi.


?BIN$(2)

10

?BIN$(10)

1010


Alama "` "(kibonye SHIFT + "@" au kushoto kwa "1") inabadili namba jozi kuwa
namba.


?`1111

15

?`100000000

256


0, 1, 2, 3, 4 ... ihesabu kwa namba jozi.

FOR I=0 TO 256:?BIN$(I),I:WAIT 30:NEXT

0  0 

1  1 

10  2 

11  3 


Namba jozi zinahesabu kwa biti "bit". 

3 ni biti 2 zenye "11", 8 ni biti 4 zenye "1000".
Jaribu sasa!

1. Hebu fanya "16" kuwa namba jozi

2. Hebu fanya namba jozi "101", "1010", "10100" kuwa namba

3. Hebu fanya "40" kuwa namba jozi

4. Ni namba gani ya kiwango cha juu ambayo inaweza kuwakilishwa na biti 5?

5. Hebu fanya "-1" kuwa namba jozi

Hifadhi namba / LET
Kompyuta ambayo ina kumbukumbu mashuhuri, inaweza kuhifadhi namba katika
muundo-wa-herufi A mpaka Z!

LET A,1190:LET B,2014:LET C,794


Endapo itaonyesha "OK", hifadhi imekamilika!
Futa skrini, na angalia kama zilihifadhiwa A? B? C?

CLS

?A

?B

?C


Inatumia "LET" kwa mara nyingine, kupandana kwa kumbukumbu.

LET B,2017

?B


Inaweza kutumia muundo-wa-herufi badala ya namba.

LET C,A+B

?C

LET C,C+1

?C


"="(Sawasawa) Alama hii inaweza kutumika kuandika kwa kifupi.

C=C+1


Sauti ambayo inapungua hatua kwa hatua!

10 LET A,1

20 ?A:BEEP A:WAIT 10:LET A,A+1

30 IF A=10 END

40 GOTO 20


Jaribu sasa!

1. Hebu tengeneza sauti ambayo inazidi kupungua

2. Hebu tengeneza sauti ambayo inaongezeka hatua kwa hatua

3. Ifanye ihifadhi mwaka wa kuzaliwa, mwezi, siku, na tunza namba ambayo
imejumlisha vyote

4. Hebu hifadhi na onyesha namba inayozidisha mwezi wa kuzaliwa na siku

5. Hebu jaribu ni tarakimu ngapi zinaweza kuhifadhiwa kwa wingi iwezekanavyo.

Unganisha programu / LRUN
IchigoJam inaweza kutunza hadi programu nne.

Unganisha programu moja na nyingine, ili iwe ya ustadi sana!

Kwanza, tengeneza mchezo wa chemsha bongo, sehemu kuu ya mchezo.

10 CLS

20 LC10,10:INPUT "1+1=?",N

30 LC12,10:IF N=2 ?"SAFI!!" ELSE ?"SHINDANO LIME
KWISHA!"


Itaonyesha "safi" kama ingizo la jibu ni sahihi! Kama sio itaonyesha "shindano
limekwisha"!.

Tunza hili katika faili namba 1.

SAVE1


Ili kuumba kichwa cha programu, awali ya yote tokea mwanzo.

NEW


Kichwa cha programu kufikia mwisho endapo kibonye chochote kitabonyezwa.

10 CLS

20 LC6,7:?"=CHEMSHA BONGO HESABU ZA KIINIMACHO="

30 WAIT 60

40 LC6,18:?"GONGA KIBONYE CHOCHOTE ILI KUANZA"

50 IF INKEY()=0 CONT


Ongeza komandi iliyounganishwa "LRUN".

60 LRUN1

SAVE0

RUN


Imeunganishwa katika mchezo kutokea kwenye kichwa cha programu!
Jaribu sasa!

1. Hebu fanya iwe jaribu-tena/mchezo-umekwisha kwa "LRUN0"

2. Hebu umba upya kichwa cha programu

3. Hebu umba upya chemsha bongo

4. Hebu ifanye irudi kwenye skrini yenye kichwa cha programu kama mchezo
umekwisha

5. Hebu weka kichwa katika "Kawakudari Game"

Julisha mawazo / BTN
Hebu iambie kompyuta mawazo yako kwa kutumia kitufe cha upande wa kushoto
katika IchigoJam.

?BTN()


Kama ukibonyeza kibonye ENTER bila kubonyeza kitufe, itaonyesha sifuri (0).
Bonyeza kishale cha juu mara tatu kilicho kwa chini kulia katika kibodi, sogeza
kishale cha mraba kinachopepesa, halafu shindilia chini kitufe katika ichigojam na
bonyeza kibonye ENTER.

Ni mafanikio kama itaonyesha "1"!
Rudia kwa kutumia komandi ya "GOTO".

10 ?BTN()

20 GOTO 10

RUN


"0" itaonyesha katika safu, hebu tujaribu kubonyeza au kuachia kitufe.

Kuisimamisha, tumia kibonye ESC(Escape).
Programu ambayo LED inawaka endapo kitufe kitabonyezwa.

10 LED BTN()

20 GOTO 10


Pia unaweza kutumia kibonye F5 badala ya komandi ya "RUN"!

Komandi ya "LED" itawasha endapo ni 1 na itapotea endapo ni 0, hivyo inawaka
shanjari na kitufe kinavyobonyezwa.

Simamisha kwa kibonye ESC(Escape).
Ukibonyeza kitufe, ikonyeze haraka!

10 LED 1:WAIT 10-BTN()*5

20 LED 0:WAIT 10-BTN()*5

30 GOTO 10


"WAIT" ni komandi ya kusubiri.

Wakati ambapo kitufe hakijabonyezwa, ni "WAIT 10".

Kama kikibonyezwa, BTN() itakuwa 1, toa namba inayozidishwa na 5 kwa "* 5",
hivyo inakuwa "WAIT 5".

Rudi nyuma katika mstari wa 10 kwa "GOTO", rudia.
Jaribu sasa!

1. Hebu tufanye ipepese kwa haraka zaidi, endapo kitufe kitabonyezwa.

2. Hebu tufanye ipepese taratibu, endapo kitufe kitabonyezwa.

3. Hebu tuiumbe upya iweze kuwasha wakati wote, endapo kitufe kitabonyezwa.

4. Hebu tuiumbe upya isiweze kuwasha kabisa, endapo kitufe hakitabonyezwa.

Tengeneza kete au dadu / RND (RANDOM)
Hebu tutengeneza kete mbalimbali!

?RND(3)


Namba ya aina gani imejitokeza?

RND (random) ni kete ambayo kuna nyuso zenye idadi sawa na namba iliyo
katika mabano.
Chapa tena na bonyeza kibonye ENTER!

?RND(3)


Je namba iliyoonyeshwa imebadilika kulinganisha na mwanzo?

Oh! Lakini 3 haikutokea?

Hii inamaanisha kwamba namba tatu na 0 ikijumuishwa zitatokea.

Kama namba katika mabano ni 3, Ni namba 3 ambazo ni 0, 1, 2.

?RND(5)


Unaweza pia kubadili namba iliyo katika mabano.
Unataka kuiondoa 0?

Unataka kufanya namba ndogo kabisa kuwa ni 1 katika namba itakayotokea?

Katika jambo kama hilo ...

?RND(3)+1


Hii ni SAWA!

Kwa kuongeza 1 katika 0 ambayo ndio ilikuwa namba ndogo zaidi, namba ndogo
zaidi katika namba itakayotokea inakuwa 1.


?RND(3)+3


Inaweza pia kubadili namba ndogo zaidi kati ya namba zitakazotokea.
Jaribu sasa!

1. Hebu tuonyeshe kwa nasibu namba yoyote kutoka 0 mpaka 6

2. Hebu tuonyeshe kwa nasibu namba yoyote kutoka 1 mpaka 6

3. Hebu tuonyeshe kwa nasibu namba yoyote kutoka 10 mpaka 20

4. Hebu tutengeneze kete ambayo itatoa namba uipendayo


CC BY IchigoJam print https://ichigojam.github.io/print/sw/ IchigoJam®jig.jp

https://creativecommons.org/licenses/by/4.0/deed.ja
https://ichigojam.github.io/print/sw/
http://ichigojam.net/


Tengeneza kibambo asili / PCG
"Kinyau" kitajitokeza kama utashikilia kibonye ALT cha kushoto katika kibodi na
kubonyeza "C".

?CHR$(236)


Msimbo wa kibambo wa "kinyau" ni 236.
Ili kubadili hiki "kinyau" kuwa "fimbo".

POKE 236*8,8,8,8,8,8,8,8,8


Imebadilika!?

Sasa ni muda wa kubadili kuwa kimpira kidogo.

POKE 236*8,0,0,0,24,24,0,0,0


Kuwa mstari mshazari!

POKE 236*8,1,2,4,8,16,32,64,128


Programu ya kuandika mchoro uupendao kwa kutumia namba jozi, 0 na 1.

10 POKE236*8+0,`01111110

11 POKE236*8+1,`11011011

12 POKE236*8+2,`11111111

13 POKE236*8+3,`11000011

14 POKE236*8+4,`11111111

15 POKE236*8+5,`11111111

16 POKE236*8+6,`00011000

17 POKE236*8+7,`00011000

RUN


Bwana barafu anatokea!
Kurudi kwa "kinyau", tumia "CLP" (kifuta ruwaza).

CLP


Jaribu sasa!

1. Hebu tengeneza alama ya moyo

2. Hebu tengeneza kibambo ukipendacho

3. Hebu tuandike upya viumbe vigeni katika ALT + D kwa kubadili 236 kuwa 237
(masafa ya msimbo wa kibambo ni 224-255!)

4. Hebu tuonyeshe namba kwa ajili ya namba jozi "10000000"

5. Hebu tutengeneze msimbo wa kibambo asili kwa programu ya mstari mmoja .

Tengeneza saa ya kupima wakati / TICK
1,2,3,4,5,6...

mara 60 kama ni sekunde 1, itaendelea kuhesabu muda, IchigoJam.

?TICK()


"TICK" ni hiyo namba. Hebu tufanye ionyeshe mara kadhaa.

?TICK()


Namba iliyoonyeshwa kwa mara ya pili ni kubwa kuliko mara ya kwanza.

?TICK()/60


Kwa kugawanya kwa 60, itabadili muda tokea ulipowasha kuwa katika mfumo wa
sekunde.

Tumia komandi "CLT" (kifuta wakati), kuweka upya (reset) namba ambayo
inahesabu.

CLT

OK

?TICK()/60


Unaweza kufahamu muda tangu kuwekwa upya (reset).
Saa ya kupima wakati ambayo inapima muda kwa sekunde tangu ilipoanzishwa
komandi ya "RUN" hadi kitakapobonyezwa kitufe cha IchigoJam.

10 CLT

20 IF BTN()=0 GOTO 20

30 ?TICK()/60


Imekuwa ikisubiri kwa kurudiarudia mpaka ambapo kitufe kilibonyezwa katika
mstari wa 20.
Kutengeneza saa ya kupima wakati ambayo itaonyesha muda kila mara
utakapobonyeza kitufe.

40 IF BTN()=1 GOTO 40

50 GOTO 10


Jaribu sasa!

1. Hebu vuta pumzi kubwa, na pima ni sekunde ngapi unaweza kutweta kwa ajili
ya pumzi.

2. Hebu funga macho, na bonyeza kitufe ili kupata tofauti ya sekunde 5.

3. Hebu tufute "/ 60" katika mstari wa 30 na pima kasi yako mwenyewe ya
kubonyeza kitufe tena na tena.

4. Tunza kwa "SAVE0", zima, washa wakati umeshikilia kitufe, kama ilianza
otomatiki, hebu bonyeza tena na tena!

5. Hebu tengeneza upya ili ianze kwa kubonyeza kitufe, na kuonyesha kwa
kubonyeza kitufe kwa mara ya pili.

1 + 1 = 1, Fumbo la Mantiki
Tuna 0 na 1 tu (katika ulimwengu wa mantiki), sasa tufanye ukokotozi. Mosi
tunaanza na kujumlisha. "|" (mwambaa wima).

?1|0

1

?0|0

0

?1|1

1


Hakuna ongezeko la tarakimu kwa kujumlisha 1 na 1, bado inabakia kuwa 1, hiyo
ni "jumla ya mantiki".
Ifuatayo ni kuzidisha, tumia "&". (Tumia koma kutenganisha na kukokotoa kwa
ujumla)

?1&1,0&1,1&0,0&0

 1   0   0   0


Kama namba itazidishwa kwa 1, jibu litakuwa namba hiyo hiyo, na litakuwa 0
endapo itazidishwa kwa 0, Hiyo ni "kuzidisha ya mantiki".
Ili kujumlisha isivyo kawaida tumia "^". Hata kama namba zote mbili ni 1, jibu ni
sifuri

?1^1,0^1,1^0,0^0

 0   1   1   0

?1|1,0|1,1|0,0|0

 1   1   1   0


Ni pekee kwa 1^1, ila kwa nyinginezo zinabakia kuwa hivyo hivyo kama kwenye
"jumla ya mantiki". Hiyo ni "XOR" (exclusive OR)
Ukokotozi wa mantiki unaweza kupangwa kwa ujumla hadi kufikia 16 (nambari ya
juu zaidi)！

?BIN$(`1010|`0011,4)

1011


Kujumlisha namba jozi zenye tarakimu nne. Unaweza kukokotoa 1|0, 0|0, 1|1, 0|1
zote kwa mara moja！

Jaribu sasa！

1. Nini jawabu kwa 3|1？

2. Tumia kidokezo: ?BIN$(1),BIN$(2),BIN$(3), tatua fumbo 3|1

3. Hebu tufanye ukokotozi wa ?8323^35 na ?8323^35^35

4. Badili kuwa namba tofauti, hebu tucheze na namba za siri kwa kutumia mantiki

5. Hebu tufanye ukokotozi mbalimbali, tufanye mafumbo ya Mantiki

Jumlisha kutoka 1 mpaka 100 - IchigoJam
Kutumia "?", Uliza kompyuta "1+1".


?1+1


Ilijibu "2" kwa kibonye ENTER. Hivyo basi, vipi kuhusu "1+2"?

?1+2


Je, kuna 'bowling pins' ngapi jumla?

?1+2+3+4


Ziko 'pins' 10.

Nini kitatokea kama itaongezeka kwa safuwima 'column' 1?

Tumia kibonye kishale cha juu/kushoto/kulia, inarejeleza programu!

Chapa "+5" ili kujumlisha, na bonyeza kibonye ENTER.

?1+2+3+4+5


Yatakuwaje matokeo wakati wa kujumlisha 1 kwa 10 kama hivi?
Kujumlisha 1 kwa
10 katika mwendo huu, jibu ni ngapi?

?1+2+3+4+5+6+7+8+9+10


Vipi kama ni kujumlisha kutoka 1 mpaka 100, inaonekana kuwa gumu!?.
Hivyo tumia komandi ya kurudia, "FOR".

FOR I=1 TO 100:?I:NEXT


Andaa muundo-wa-herufi "A", itakayoweza kujumlishwa zaidi na zaidi.

A=0

OK

FOR I=1 TO 100:A=A+I:NEXT

OK

?A

5050


Jaribu sasa!

1. Yapi matokeo kwa kujumlisha toka 1 mpaka 12?

2. Yapi matokeo kwa kujumlisha toka 1 mpaka 20?

3. Yapi matokeo kwa kujumlisha toka 1 mpaka 200?

4. Yapi matokeo kwa kujumlisha toka 1 mpaka 300?

5. Ipi namba ya kikomo ambayo inaweza kukokotolewa kwa uzuri?

Kutafuta Kigawe cha Shirika(Hisabati)
Hebu onyesha namba kutoka 1 mpaka 100.

10 FOR X=1 TO 100

20 ?X

30 WAIT 3

40 NEXT


Kwa komandi ya "FOR"/"NEXT", wakati ukiongeza muundo-wa-herufi X kwa 1,
rudia mpaka ifikie 100.
Hebu onyesha kigawe cha 3 tu kwa kutumia alama "%" ambayo inakokotoa baki.

20 IF X%3=0 ?X


Wakati ikionyesha pia kigawe cha 5, tumia "OR" ikimaanisha "either".

20 IF X%3=0 OR X%5=0 ?X


Wakati ikionyesha ile ambayo ni kigawe cha 3 na 5, tumia "AND" ikimaanisha
"both".

20 IF X%3=0 AND X%5=0 ?X


Hiki ni "kigawe cha shirika" cha 3 na 5.

Namba ya kwanza, 15 ni "kigawe kidogo cha shirika".
Hebu onyesha "vigawe vya shirika" vya 2 na 3 na 5.

20 IF X%2=0 AND X%3=0 AND X%5=0 ?X


"Vigawe vya shirika" viko 3, na 30 ndio "kigawe kidogo cha shirika" kati ya 1 na
100.
Jaribu sasa!

1. Hebu onyesha vigawe vya 7

2. Hebu onyesha vigawe vya 13

3. Hebu onyesha vyote viwili vigawe vya 7 na vigawe vya 13

4. Hebu onyesha vigawe vya 7 na 13

5. Hebu kwa kasi hesabu viko vingapi vigawe vya shirika vya 3 na 7 kutoka 1
mpaka 1000

Kutafuta Kigawo cha Shirika(Hisabati)
Ni namba gani inagawanyika kwa 182?

10 N=182

20 FOR X=2 TO N-1

30 IF N%X=0 ?X

40 NEXT


Kwa komandi ya "FOR" / "NEXT", wakati tukiongeza variable X kwa 1 kutokea 2,
na kurudia.

Ni SAWA pia kwa "N=182" ikaandikwa kama "LET N, 182"!
Hebu tuangalie "kigawo", namba iliyojitokeza.

?182%91

0

?182%26

0


Inaweza kugawanya kwa uzuri!
Ni namba gani inaweza kugawanya 455?

10 N=455


Inaonyesha namba ambayo inaweza kugawanya zote mbili 182 na 455.

10 N=182:M=455

20 FOR X=2 TO N-1

30 IF N%X=0 AND M%X=0 ?X

40 NEXT


Namba zilizoonyeshwa 7, 13, na 91 ndio "vigawo vya shirika" vya 182 na 455.

Namba kubwa zaidi 91 ndio "kigawo kikubwa cha shirika" cha 182 na 455.
Jaribu sasa!

1. Hebu onyesha "kigawo cha shirika" cha 3432 na 2002.

2. Hebu gawanya 3432 kwa "kigawo kikubwa cha shirika" kutoka (1).

3. Hebu zidisha namba iliyokokotolewa hapo juu (2) kwa "kigawo kikubwa cha
shirika".

4. Hebu tuhesabu viko vingapi "vigawo" vya 2002.

5. Hebu tafuta "namba tasa" ambazo hazina "kigawo" katika namba yenye
tarakimu 4.

6. Hebu tuhakiki kwamba hakuna "kigawo cha shirika" kati ya "namba tasa"
kutoka (5).

Kukadiria namba (Hisabati)
Hebu fanya kadirio la chini!

41 ni 40, 44 ni 40, 45 pia ni 40, 49 ni 40.

Inaitwa kukadiria chini nafasi ya kwanza.

?44/10

4


Kwa kuwa ni ukokotozi wa namba kamili, kama utagawanya kwa 10, itakuwa 4
badala ya 4.4.

?44/10*10

40


Gawanya kwa 10, na zidisha kwa 10, imemaliza kadirio la chini!

Hebu tufanye kadirio la juu!

49 ni 50, 45 ni 50, 44 & 41 pia ni 50, 40 ni 40.

Inaitwa kukadiria juu nafasi ya kwanza.

?44/10*10+10

50


Baada ya kukadiria chini, unaweza kuikadiria juu kwa kujumlishia 10!.

?40/10*10+10

50


Oh safi, 40 imebadilika na kuwa 50. Hebu amua kwa nafasi ya kwanza "ones
place".

10 INPUT N

20 LET M,N%10

30 ?M


"%" ni alama ya kukokotoa baki katika ulimwengu wa IchigoJam BASIC.

Kama nafasi ya kwanza ni 0, ikadirie chini, na kama ni 1 au zaidi ikadirie juu.

40 IF M=0 ?N/10*10

50 IF M>=1 ?N/10*10+10


Jaribu sasa!

1. Hebu kadiria juu nafasi ya kwanza ya 72 kwa kutumia programu.

2. Hebu tengeneza programu ya "kukadiria namba" ambayo itakadiria chini nafasi
ya kwanza kama ni 4 au chini yake na ikadirie juu kama ni 5 au zaidi yake.

3. Hebu tuhakiki kama inaweza kukadiria juu nafasi ya kwanza hata kwa "?
(N+9)/10*10".

4. Hebu tengeneza programu ya kukadiria namba nafasi ya kwanza.

ChemshaBongo ya Namba
Hizi ni namba kwa Kiswahili, lugha ya Tanzania.

10 LET[0],"SIFURI","MOJA","MBILI","TATU","NN
E","TANO","SITA","SABA","NANE","TISA","KUMI"

RUN


Jinsi ya kuonyesha namba kwa neno.

?STR$([1])


Iache kompyuta yako ihesabu!

20 FOR I=10 TO 0 STEP -1

30 ?I,STR$([I]):WAIT60

40 NEXT


Mchezo wa 1: Jibu mchezo wa namba!

20 N=RND(11):?STR$([N]);

30 INPUT ". What's this?",M

40 IF N=M ?"Correct!" ELSE ?"Wrong. The answer i
s ";N

50 GOTO 20


Mchezo wa 2: Chapa ili kujibu mchezo wa namba!

20 N=RND(11):?"Type ";N

30 FOR I=0 TO LEN([N])-1

40 IF INKEY()!=ASC([N]+I) CONT

50 ?STR$([N]+I,1);:NEXT:?

60 GOTO 20


Jaribu sasa!

1. Ongeza "MCHEZO-UMEKWISHA" katika mchezo wa 1 kama si sahihi

2. Ongeza alama/maksi katika mchezo wa 1

3. Onyesha muda baada ya majibu 10 katika mchezo wa 1

4. Ongeza "MCHEZO-UMEKWISHA" katika mchezo wa 2 kama ukikosea
kuchapa

5. Badili lugha kuwa nyingine

Mchezo wa Mkono (Janken)
Mchezo wa mkono (maarufu kama Rock-Paper-Scissor) ni mchezo unaotumika
haswa kuchagua bila upendeleo baina ya timu fulani.Kwa kutumia mkono kuna
ishara tatu rock(jiwe), paper(karatasi) na scissors(mkasi)

- Gu: (rock) ni kukunja ngumi. (Gu inamshinda Choki)

- Choki: (scissors) tumia vidole viwili, igiza umbo la mkasi. (scissors inamshinda
Pa)

- Pa: (paper) nyosha chini kiganja chako. (Pa inamshinda Gu)

Tumia komandi ya "Random" yenye namba 3 kama unataka kuamua Gu, Choki
au Pa kwa kompyuta yako.

?RND(3)


Kwa mfano 0 inamaanisha Gu, 1 inamaanisha Choki, 2 inamaanisha Pa.

10 LET A,RND(3)

20 IF A=0 ?"GU"

30 IF A=1 ?"CHOKI"

40 IF A=2 ?"PA"

RUN


Jaribu kuchapa RUN au bonyeza kibonye F5 kwa kurudia rudia!

Bonyeza kibonye F5 na sema alama ya mkono wako ili kushindana baada ya
kutamka "Jan ken"!

Jaribu kuamua kwa kompyuta.

15 INPUT "YOUR HAND?",B

16 IF B=0 ?"GU - ";

17 IF B=1 ?"CHOKI - ";

18 IF B=2 ?"PA - ";

50 IF A=B ?"DRAW"

60 IF A-B=1 OR B-A=2 ?"WON!"

70 IF B-A=1 OR A-B=2 ?"LOST!"


Jaribu sasa!

1. Jaribu kushindana na programu katika kompyuta ya rafiki yako!

2. Jaribu kuumba kompyuta dhaifu kabisa!

3. Jaribu kuumba mchezo wa Janken wenye wachezaji 3, wewe na kompyuta 2!

Mchezo wa Kawakudari
"Mchezo wa Kawakudari" ni mchezo ambao unaepa vizuizi na kutembea katika
mto.
Kwanza, amua kibambo chako mwenyewe.

10 CLS:X=15

20 LC X,5:?"O"

RUN


※Kwa 「"」, shikilia kibonye Shift na bonyeza 「'」. Kwa 「?」, shikilia kibonye
Shift na bonyeza 「/」. Wakati mwingine maeneo haya yanaweza kubadilika
kutegemeana na aina ya kibodi.
Kibambo cha adui, kujitokeza.

30 LC RND(32),23:?"*"


※ Kwa 「(」「)」, shikilia kibonye Shift na bonyeza 「8」「9」. 

Hebu bonyeza kibonye F5(sawa kama "RUN") mara nyingi.
Rudia katika programu.

40 GOTO 20


Inakwenda kasi sana? Simamisha kwa kibonye ESCAPE, na ingiza "WAIT".

35 WAIT 3


Kuweza kusogea kwa kibonye mshale cha kushoto na/au kulia .

36 X=X-BTN(28)+BTN(29)


Pachika maamuzi ya kugonga ili kumalizika kwa mchezo!

37 IF SCR(X,5) END


Kama hujagongwa, mchezo unaendelea.
Jaribu sasa!

1. Hebu rekebisha mwendo

2. Hebu badili kuwa kibambo chako mwenyewe (shikilia kibonye ALT na bonyeza
moja kati ya A mpaka V)

3. Hebu tuongeze vizuizi(mf. "***")

4. Hebu tufikiri kuhusu hadithi ya mchezo

5. Hebu tafuta vidokezo/ujanja(virusi)

6. Hebu fikiri jinsi ya kufuta kirusi

Gusa Kuandika
Gusa kuandika ili kubonyeza bila kuangalia katika kibodi, ni wa kasi! burudani!
poa!


10 S="JKL"

20 CLT:FOR I=1 TO 10

30 N=RND(LEN(S)):C=ASC(S+N)

40 ?CHR$(C)

50 K=INKEY():IF K=0 CONT

60 IF C<>K ?"MISS!":END

70 NEXT:?TICK()/60


Weka kidole cha shahada cha mkono wa kulia katika nafasi ya "J", kidole cha kati
katika "K", kidole cha pete katika "L".

Chapa haraka haraka J, K au L ambayo inaonyeshwa bila mpangilio maalum.

Je unaweza kupiga mara 10 bila kukosea? Lengo ni sekunde 8!
Baada ya kuifuta, hebu tuongeze vibambo alfabet hatua kwa hatua.

Weka kidole cha shahada cha mkono wa kushoto katika "F".

Unaweza kupata rekodi ya sekunde-8 kwa "FJKL"?

10 S="FJKL"


Kidole cha kati cha mkono wa kushoto katika "D", kidole cha pete katika "S" na
kidole kidogo katika "A".

Hii ndio "home position" kwa kugusa kuandika!

10 S="ASDFJKL"


Jaribu sasa!

1. Hebu tupatie rekodi ya sekunde 8 kwa "ASDFJKL"

2. Hebu ongeza "H" ambayo inabonyezwa na kidole cha shahada cha mkono wa
kulia

3. Hebu ongeza "G" ambayo inabonyezwa na kidole cha shahada cha mkono wa
kushoto

4. Hebu ongeza "ZXCVBNM" iliyopo kwa chini katika kibodi

5. Hebu ongeza "QWERTYUIOP" iliyopo kwa juu katika kibodi

6. Hebu fikia rekodi ya sekunde-8 kwa alfabeti zote

Q W E RT YU I O P 

A S D FG HJ K L ; 

Z X C VB NM , . / 


Jedwali la ugawaji wa vibonye kwa kila kidole

Mchezo wa kuchapa Namba
Mchezo wa kuchapa Jina(chapa jina lako katika mstari wa 20!)

10 CLS

20 ?"TAISUKE FUKUNO"

30 CLT:P=0

40 C=SCR(P,0)

50 IF C=0 ?:?TICK()/60:END

60 IF INKEY()<>C CONT

70 ?CHR$(C);:P=P+1

80 GOTO 40


Tayari umezoea?

Umba upya kuwa mchezo wa kuchapa alfabet ABC!

10 FOR I=1 TO 10:?CHR$(65+RND(3));:NEXT


Unaweza kukatisha katika sekunde ya 20??
Umba upya kuwa mchezo wa kuchapa namba!

20 FOR I=2 TO 5:?RND(256);",";:NEXT:?RND(256)


Unaweza kukatisha katika sekunde ya 10?
Kivipi alama zinaweza kuonyeshwa hadi katika nafasi ya kwanza ya desimali?

T=TICK():?T/60;".";T%60/6


Kwa sekunde 0.1, shindana!
Jaribu sasa!

1. Hebu umba upya kuwa mchezo uupendao wa kuchapa maneno

2. Hebu tujifunze maneno magumu ya Kiingereza kwa mchezo wa kuchapa

3. Hebu umba upya kuwa mchezo wa namba jozi

4. Hebu umba upya kuwa mchezo wa kuchapa alfabeti bila mpangilio

5. Hebu umba upya kuwa mchezo wa kuchapa emoji

Mchezo wa Kuruka
Mchezo wa kuruka ambao unamuepa adui kwa kuruka.

Kwanza, kibambo kikuu kujitokeza!

10 CLS:Y=15:V=0

20 LC 0,Y:?"M"


Kama utabonyeza kibonye chochote, kibambo kikuu kitaruka!

30 IF Y=15 V=3*(INKEY()!=0)

40 Y=Y-V:V=V-1

50 SCROLL LEFT

60 GOTO 20


Kuweka vibambo vya maadui.

25 IF RND(10)=0 LC 31,15:?"K"


Rekebisha kasi.

35 WAIT 4


Ongeza maamuzi ya mchezo-umekwisha, Umemalizika!

37 IF SCR(1,Y) END


Jaribu sasa!

1. Hebu tubadili kibambo kikuu kuwa "R"

2. Badili kibambo kikuu, shikilia kibonye ALT na bonyeza "R"

3. Hebu badili "RND(10)" kuwa "RND(4)"

4. Hebu punguza vibambo vya adui

5. Hebu fanya kasi iwe haraka mno

6. Hebu badili "V=3*" kuwa "V=4*"

7. Hebu punguza nguvu ya kuruka

8. Hebu ongeza alama 'score' kwa kutumia "CLT" na "TICK()"

9. Hebu hesabu na weka alama ya idadi uliyoruka

10. Hebu tengeneza mchezo wa asili kwa kuumba upya

Mchezo wa Nekonikoban
Maana ya "Neko ni koban" inakaribiana na "Lulu kwa nguruwe"(upotezaji mkubwa
wa rasilimali). "Neko"- "Nyau", "Koban"-"Sarafu"
Nyau atajitokeza katika nafasi ya
10 toka kushoto na 5 toka juu.(msimbo wa kibambo ni 236, ALT + C)

10 CLS:X=10:Y=5

50 LC X,Y:?CHR$(236)


Ifanye iweze kusogea kwa kielekezi.

20 K=INKEY()

30 X=X-(K=28)+(K=29)

40 Y=Y-(K=30)+(K=31)

60 WAIT10

70 LCX,Y:?" "

80 GOTO20


Tawanya visarafu kwa uhuru, na ongeza alama/score(S).

15 S=0:FOR I=1 TO 10:LOCATE RND(30),RND(20):?CHR
$(245):NEXT

45 IF SCR(X,Y)=245 S=S+1:LC0,0:?S


Mchezo unakwisha ndani ya sekunde 10.

7 CLT

47 IF TICK()>600 END


Jaribu sasa!

1. Hebu onyesha "LIST"

2. Hebu toa idadi kubwa zaidi ya visarafu

3. Hebu tega iwe mchezo-umekwisha ndani ya sekunde 5

4. Hebu iwezeshe kusogea kwa haraka

5. Hebu tufanye alama ziongezeke kwa pointi 10

6. Hebu rekebisha na hifadhi vile upendavyo

Population simulation
Idadi ya Wajapani ni Milioni 125. Ilipunguwa kwa ndani ya mwaka mmoja270,000.

(Nov, 2016 Wizara ya Mambo ya Ndani na Mawasiliano, Ofisi ya Takwimu,
Uhakikisho wa idadi ya watu)

10 P=12536

20 D=27


Seti idadi ya watu kama "P", na namba inayopungua kama "D", kwa zote uniti
iwakilishwe na elfu kumi.

Endapo ikiendelea kupungua katika hali sawa sawa kila mwaka tokea 2016, Idadi
ya watu itakuwaje ifikapo 2100?

30 FOR Y=2016 TO 2100

40  ?Y,P;

50  ?

60  P=P-D

70 NEXT


Hebu tuifanye kuwa jedwali ('/' ni alama ya kugawanya).

42  FOR I=0 TO P/1000

45  ?"*";:NEXT


Vipi kama namba inayopungua ikaongezeka kwa 20,000 kila mwaka?

55  D=D+2


Jaribu sasa!

1. Simamisha kwa kibonye ESC na angalia idadi ya watu kwa mwaka 2060

2. Hebu fanya namba inayopungua kuwa 30,000 kila mwaka

3. Hebu badili kibambo "*" kuwa kibambo kingine

4. Hebu badili namba katika mstari wa 42, 1000 kuwa 100
5. Hebu ongeza idadi ya watu

6. Idadi ya watu duniani ni Bilioni 7.3. Itakuwaje kama ongezeko ni Milioni 100 kila
mwaka?

Siku gani katika wiki ulizaliwa?
Ingizo Mwaka, mwezi, tarehe, na fahamu ni siku gani katika wiki?

Kwa hakika, inaweza kukokotolewa kwa programu.

Kwanza, programu ya kuonyesha siku za wiki.

10 LET[0],"SUN","MON","TUE","WED","THU","FRI","S
AT"

RUN

?STR$([0])


Tumia "STR$" kuonyesha mwanzo [0] wa tungo"string" ndani ya
mkusanyiko"array".
Rudia kutoka Sunday (Jumapili) mpaka Saturday (Jumamosi) kwa siku 7 za wiki.

FOR I=0 TO 22:?STR$([I%7]):NEXT


Tumia "INPUT", weka mwaka katika "Y", mwezi katika "M" na tarehe katika "D".

20 INPUT"Y?",Y

30 INPUT"M?",M

40 INPUT"D?",D

RUN


Hebu onyesha maingizo ya mwaka, mwezi na tarehe.

?Y,M,D


"Kanuni ya Zeller" ambayo inakokotoa siku ya wiki kutokana na mwaka, mwezi na
tarehe.


50 IF M<3 M=M+12:Y=Y-1

60 H=(Y+Y/4-Y/100+Y/400+(13*M+8)/5+D)%7

70 ?STR$([H])


Ni ya kushangaza kuweza kujua siku katika wiki kwa hii!
Jaribu sasa!

1. Weka tarehe ya leo na angalia leo ni siku gani

2. Siku yangu ya kuzaliwa ni siku gani katika wiki?

3. Siku yangu ya kuzaliwa inayofuata ni siku gani katika wiki?

4. Hebu onyesha siku za wiki kwa lugha nyingine

5. Hebu tengeneza kalenda ya miezi ya kuzaliwa

Gacha simulation
Hebu malizia alfabeti kwa gacha!

10 A=RND(26)

20 ?"DRAW:";CHR$(ASC("A")+A)


Alfabeti 26 kutoka A mpaka Z zinatumika kucheza gacha!. Tumia "RND" ili
kucheza.
Hebu onyesha mkusanyiko.

30 [A]=[A]+1

40 FOR I=0 TO 25

50  LC I%6*5,5+I/6*3

60  ?CHR$(ASC("A")+I);":";[I]

70 NEXT


Mpaka imalizike, gacha otomatiki!

5 CLS:CLV

35 N=N+1:C=0

65  IF [I] C=C+1

80 ?:?:?"COLLECT:";C;"/26 ";

90 ?N;"00YEN"

100 IF C=26 ?"COMPLETE!":END

110 GOTO 10


Itagharimu kiasi gani?
Hebu fanya "A" kuwa adimu.

15 IF A=0 IF RND(10)>0 A=1


Ifanye 'probability' yake ya kuwa "B" iwe 9/10 hata kama "A" itajitokeza.
Jaribu sasa!

1. Hebu jaribu mara tatu, na linganisha

2. Hebu fanya kuwa adimu kutoka "A" mpaka "F"

3. Hebu tuhakikishe kuwa "Z" haijitokezi

4. Hebu tuitoe "Z" kwa uwezekano wa 1/2

5. Hebu tuifanye ipandiane (ziwe tofauti)

